


How to add Zenon 7.20 import to Nimbus

Zenon 7.20 is able to send alarm to Nimbus using a *Generic Text File* and a VBA script.



In your Zenon Supervisor, open the VBA Editor

```

Public x5monitoringControl As x5monitoring
Dim WithEvents x5pic As zenOn.DynPicture
Dim WithEvents zAlarm As Alarm

Private Sub Project_Active()
    Set x5pic = thisProject.DynPictures.Item("Automotive_Stratonmonitoring")
    Set zAlarm = thisProject.Alarm
End Sub

Private Sub Project_Inactive()
    If Not x5monitoringControl Is Nothing Then
        Set x5monitoringControl = Nothing
    End If
    Set zAlarm = Nothing
End Sub

Private Sub x5pic_Close()
    If Not x5monitoringControl Is Nothing Then
        Set x5monitoringControl = Nothing
    End If
End Sub

Private Sub zAlarm_AlarmComes(ByVal obItem As IAlarmItem)
    Alarm2Nimbus obItem, 1
End Sub
Private Sub zAlarm_AlarmGoes(ByVal obItem As IAlarmItem)
    Alarm2Nimbus obItem, 0
End Sub
Private Sub zAlarm_AlarmAcknowledged(ByVal obItem As IAlarmItem)
    Alarm2Nimbus obItem, 2
End Sub
Private Sub Alarm2Nimbus(ByVal obItem As IAlarmItem, iStatus As Integer)
    Dim sNimbusFile As String
    Dim sLine As String
    Dim iLineNumber As Integer
    '
    ' Should reflect same file name as entered in Nimbus Explorer ->
    ' Setup -> SCADA Import -> Generic File
    '
    sNimbusFile = "C:\ProgramData\TroSoft\Nimbus Alarm Server 3\Project\Zenon.txt"

    On Error Resume Next

    iLineNumber = FreeFile

    Open sNimbusFile For Append As #iLineNumber

    If Err.Number = 0 Then

```

In the *VBA Editor*, insert code as above. Mainly the code contains an *Alarm* object with events that triggers three functions (methods): *zAlarm_AlarmComes*, *zAlarm_AlarmGoes* and *zAlarm_AlarmAcknowledged*. These functions calls a function, *Alarm2Nimbus*, which writes the alarm event and its info to a text file. The *Alarm* object (*zAlarm*) must be created in the *Project_Active* function and destroyed in the *Project_Inactive* function.


You will find the code sections later in this document. Copy and paste the code to the appropriate places in your own VBA code or create a new VBA code.

Change the *sNimbusFile* variable to point to the folder where you wish the Alarm export file. In the example it points to the Nimbus 3 default project folder.


Start your *Zenon* project and create and alarm. You will see the *sNimbusFile* file be created.

Now it's time to configure Nimbus

Start *Nimbus Explorer* (right click and 'Run as Administrator') from the start button menu shortcut. Actually *Nimbus Explorer* should always be run as *Administrator* by selecting this option in the shortcut.



Select *Setup* -> *SCADA import Setup*.



Select *SCADA System* -> *Add SCADA System Import* -> *Generic File*

Now select the *Zenon.txt* file in the same directory as pointed by the *sNimbusFile*. If the file does not exist yet select another file in the same directory and change the path/name manually.

Obs! It is important not to leave the wrong file name there as *Nimbus* will try to delete the file when it starts up!

When you start *Nimbus Alarm Server* it will remove the file, because it contains only old alarm events.

Filter profile

Save Import Export Delete

Variable name	Value	Measuring unit	Text	User - full name
Energy_windfarm[24].Alarm	2		Windturbine Alarm (Level2)	
Energy_Hydro_Process_WaterLevelUpstream	937	m ü.M.	Water Level High Warning	
Energy_Hydro_Process_WaterLevelUpstream	865	m ü.M.	Water Level LOW Warning	
Energy_Hydro_Process_WaterLevelUpstream	861	m ü.M.	Water Level LOW Alarm	SYSTEM
Energy_Hydro_Process_WaterLevelUpstream	861	m ü.M.	Water Level LOW Warning	

Some alarms from *the Zenon Demo Project* as they appear using the *Nimbus Popup* form.

The VBA Code which does the job

```

Dim WithEvents zAlarm As Alarm

Private Sub Project_Active()

    Set zAlarm = thisProject.Alarm

End Sub

Private Sub Project_Inactive()

    Set zAlarm = Nothing

End Sub

Private Sub zAlarm_AlarmComes(ByVal obItem As IAlarmItem)

    Alarm2Nimbus obItem, 1

End Sub

Private Sub zAlarm_AlarmGoes(ByVal obItem As IAlarmItem)

    Alarm2Nimbus obItem, 0

End Sub

Private Sub zAlarm_AlarmAcknowledged(ByVal obItem As IAlarmItem)

    Alarm2Nimbus obItem, 2
  
```

```

End Sub

Private Sub Alarm2Nimbus(ByVal obItem As IAlarmItem, iStatus As Integer)

    Dim sNimbusFile As String
    Dim sLine As String

    Dim iFileNumber As Integer

    '
    ' Should reflect same file name as entered in Nimbus Explorer ->
    ' Setup -> SCADA Import -> Generic File
    '
    sNimbusFile = "C:\ProgramData\TroSoft\Nimbus Alarm Server 3\Project\Zenon.txt"

    On Error Resume Next

    iFileNumber = FreeFile

    Open sNimbusFile For Append As #iFileNumber

    If Err.Number = 0 Then

        sLine = CStr(iStatus) & "|"

        Select Case iStatus
            ' Goes
            Case 0:
                sLine = sLine & Format(obItem.Timegoes, "yyyy-mm-dd|Hh:Nn:Ss|")
            ' Comes
            Case 1:
                sLine = sLine & Format(obItem.Timecomes, "yyyy-mm-dd|Hh:Nn:Ss|")
            ' Acknowledged
            Case 2:
                sLine = sLine & Format(obItem.Timeacknowledged, "yyyy-mm-dd|Hh:Nn:Ss|")
            Case Else
                sLine = sLine & Format(Now, "yyyy-mm-dd|Hh:Nn:Ss|")
        End Select

        sLine = sLine & obItem.Name & "|" & CStr(obItem.AlarmGroup) & "|"
        sLine = sLine & CStr(obItem.AlarmClass) & "|" & "" & "|" & obItem.Text & " "
        sLine = sLine & obItem.Userstext

        Print #iFileNumber, sLine

    End If

    Close #iFileNumber

    On Error GoTo 0

End Sub

```